
Abstract. The TPTP World is a well-established infrastructure for automatic
theorem provers. It defines several concrete syntaxes, notably an untyped first-
order form (FOF) and a typed first-order form (TFF0), that have become de facto
standards in the automated reasoning community. This paper introduces the TFF1
format, an extension of TFF0 with rank-1 polymorphism. It presents its syntax,
typing rules, and semantics, as well as a sound and complete translation to TFF0.
The format is designed to be easy to process by existing reasoning tools that sup-
port ML-style polymorphism. It opens the door to useful middleware, such as
monomorphizers and other translation tools that encode polymorphism in FOF
or TFF0. Ultimately, the hope is that TFF1 will be implemented in popular auto-
matic theorem provers.

1 Introduction

The TPTP World [28] is a well-established infrastructure for supporting research, devel-
opment, and deployment of automated reasoning tools. It owes its name to its vast prob-
lem library, the Thousands of Problems for Theorem Provers (TPTP) [27]. In addition,
it specifies concrete syntaxes for problems and solutions: Dozens of reasoning tools
implement the TPTP untyped clause normal form (CNF) and first-order form (FOF) for
classical first-order logic with equality.

It has often been argued that the gap between the features supported by provers and
those needed by applications is too wide, and that rich interchange formats are needed to
address this disconnect [16, 18, 23, 25, 32]. A growing number of reasoners can process
the recently introduced TPTP “core” typed first-order form (TFF0) [30], with mono-
morphic types and interpreted arithmetic [15, 22], or the corresponding higher-order
form (THF0) [2]; a polymorphic version of THF0, the full THF, is in the works [29].

Despite the variety of this offering, there is a strong desire in part of the automated
reasoning community for a portable polymorphic first-order format. Many applications
require polymorphism, notably interactive theorem provers and program specification
languages; but lacking a suitable syntax, applications and provers must communicate
via monomorphic formats. To make matters worse, there is no entirely satisfactory
way to eliminate polymorphism: Monomorphization algorithms are necessarily incom-
plete [7, §2], and it is difficult to encode polymorphism in a complete yet also sound
and efficient manner, especially in the presence of interpreted types. Tool authors are



reduced to developing their own monomorphizers and type encodings, often using sub-
optimal schemes. Polymorphism arguably belongs in provers, where it can be imple-
mented simply and efficiently, as demonstrated by the SMT (satisfiability modulo the-
ories) solver Alt-Ergo [6].

This paper introduces the TFF1 format, an extension of TFF0 with rank-1 poly-
morphism. The extension was designed with the participation of members of the TPTP
community, reflecting its needs. Besides compatibility with TFF0 and conceptual in-
tegrity with the upcoming full THF, an important design goal was to ensure that the
format can easily be processed by existing reasoning tools that support ML-style poly-
morphism. TFF1 also opens the door to useful middleware, such as monomorphizers
and other tools that encode polymorphism in FOF or TFF0.

For SMT solvers, the SMT-LIB 2 format [1] specifies a classical many-sorted logic
with equality and interpreted arithmetic, much in the style of TFF0 but with parametric
symbol declarations (overloading). Polymorphism would make sense there as well, as
witnessed by Alt-Ergo. However, the SMT community is still recovering from the up-
grade to SMT-LIB 2 and busy defining a standard proof format [4]; implementers would
not welcome yet another feature at this point. Moreover, with its support for arithmetic,
TFF1 is a reasonable format to implement in an SMT solver if polymorphism is desired.

This paper is structured as follows. Section 2 specifies the TFF1 syntax, and Sec-
tion 3 specifies its typing rules and semantics. Section 4 presents a translation from
TFF1 to TFF0. Section 5 bridges the gap with ML-style formalisms by defining two
preprocessing steps that address type quantifiers in TFF1 type signatures and formu-
las. Section 6 briefly reviews the applications that already implement TFF1. Section 7
considers related work in the TPTP and SMT communities: existing polymorphic for-
malisms and translation schemes to untyped or monomorphic logics.

This specification extends, rather than replaces, the TFF0 specification [30]. The
parts that TFF1 inherits directly from TFF0, such as the concrete syntax for the logical
connectives and the (optional) arithmetic constructs, are described in more detail there.

2 Syntax

Briefly, the types, terms, and formulas of TFF1 are analogous to those of TFF0, ex-
cept that function and predicate symbols can be declared to be polymorphic, types can
contain type variables, and n-ary type constructors are allowed. Type variables in type
signatures and formulas are explicitly bound. Instances of polymorphic symbols are
specified by explicit type arguments, rather than inferred.

Types. The types of TFF1 are built from type variables and type constructors of fixed
arities. Nullary type constructors are called type constants. The usual conventions of
TPTP apply: Type variables start with an uppercase letter and type constructors with a
lowercase letter. The types A, list(A), list(bird), and map(nat, list(B)) are all
examples of well-formed types. A type is polymorphic if it contains any type variables;
otherwise, it is monomorphic.

As in TFF0, the type $i of individuals is predefined but has no fixed semantics,
whereas the arithmetic types $int, $rat, and $real are modeled by Z, Q, and R. The



TFF0 specification [30] defines the semantics of the arithmetic types and their oper-
ations. It is perfectly acceptable for a TFF implementation to restrict itself to “pure
TFFk,” without arithmetic. TFFk with arithmetic is sometimes labeled “TFAk.”

Type Signatures. Each function and predicate symbol occurring in a formula must be
associated with a type signature that specifies the types of the arguments and, for func-
tions, the result type. Type signatures can take any of the following forms:

(a) a type (predefined or user-defined);
(b) the Boolean pseudotype $o (the result “type” of predicate symbols);
(c) (τ1 * · · · * τn) > τ̃ for n > 0, where τ1, . . . , τn are types and τ̃ is a type or $o;
(d) !>[α1 : $tType, . . ., αn : $tType]: ς for n > 0, where α1, . . . ,αn are distinct type

variables and ς has one of the previous three forms.

In accordance with TFF0, the parentheses in form (c) are omitted if n = 1. The binder
!> in form (d) denotes universal quantification. If ς is of form (c), it must be enclosed
in parentheses. All type variables must be bound by a !>-binder.

Here are a few examples: (a) $int, monkey, banana; (b) $o; (c) monkey > banana,
(monkey * banana) > $o; (d) !>[A : $tType]: ((A * list(A)) > list(A)).

Form (a) is used for monomorphic constants; form (b), for propositional constants,
including the predefined symbols $true and $false; form (c), for monomorphic func-
tions and predicates; and form (d), for polymorphic functions and predicates. It is often
convenient to regard all forms above as instances of the general syntax

!>[α1 : $tType, . . ., αm : $tType]: ((τ1 * · · · * τn) > τ̃)

where m and n can be 0.
Type variables that are bound by !> without occurring in the type signature’s body

are called phantom type variables. These make it possible to specify operations and
relations directly on types and provide a convenient way to encode type classes. For
example, we can declare a polymorphic propositional constant is_linear with the type
signature !>[A : $tType]: $o and use it as a guard to restrict the axioms specifying that
a binary predicate less_eq with the type signature !>[A : $tType]: ((A * A) > $o) is
a linear order to those types that satisfy the is_linear predicate.

Type Declarations. Type constructors can optionally be declared. The following decla-
rations introduce a type constant bird, a unary type constructor list, and a binary type
constructor map:

tff(bird, type, bird: $tType).

tff(list, type, list: $tType > $tType).

tff(map, type, map: ($tType * $tType) > $tType).

If a type constructor is used before being declared, its arity is determined by the first
occurrence. Any later declaration must give it the same arity.

A declaration of a function or predicate symbol specifies its type signature. Every
type variable occurring in a type signature must be bound by a !>-binder. The following
declarations introduce a monomorphic constant pi, a polymorphic predicate is_empty,
and a pair of polymorphic functions cons and lookup:



tff(pi, type, pi: $real).

tff(is_empty, type, is_empty : !>[A : $tType]: (list(A) > $o)).

tff(cons, type,

cons : !>[A : $tType]: ((A * list(A)) > list(A))).

tff(lookup, type,

lookup : !>[A : $tType, B : $tType]: ((map(A, B) * A) > B)).

If a function or predicate symbol is used before being declared, a default type signature
is assumed: ($i * · · · * $i) > $i for functions and ($i * · · · * $i) > $o for predicates. If
a symbol is declared after its first use, the declared signature must agree with the as-
sumed signature. If a type constructor, function symbol, or predicate symbol is declared
more than once, it must be given the same type signature up to renaming of bound type
variables. All symbols share the same namespace; in particular, a type constructor can-
not have the same name as a function or predicate symbol.

Function and Predicate Application. To keep the required type inference to a minimum,
every use of a polymorphic symbol must explicitly specify the type instance. A function
or predicate symbol with a type signature

!>[α1 : $tType, . . ., αm : $tType]: ((τ1 * · · · * τn) > τ̃)

must be applied to m type arguments and n term arguments. Given the above type sig-
natures for is_empty, cons, and lookup, the term lookup($int, list(A), M, 2) and
the atom is_empty($i, cons($i, X, nil($i))) are well-formed and contain free oc-
currences of the type variable A and the term variables M and X.

In keeping with TFF1’s rank-1 polymorphic nature, type variables can only be in-
stantiated with actual types. In particular, $o, $tType, and !>-binders cannot occur in
type arguments of polymorphic symbols.

For systems that implement type inference, the following extension of TFF1 might
be useful. When a type argument of a polymorphic symbol can be inferred automati-
cally, it may be replaced with the wildcard $_. For example:

is_empty($_, cons($_, X, nil($_)))

Although nil’s type argument cannot be inferred locally from the types of its term
arguments (since there are none), the Hindley–Milner type inference can deduce it from
X’s type. The producer of a TFF1 problem must be aware of the type inference algorithm
implemented in the consumer to omit only redundant type arguments.

Type and Term Variables. Every variable in a TFF1 formula must be bound. The vari-
able’s type must be specified at binding time:

tff(bird_list_not_empty, axiom,

![B : bird, Bs : list(bird)]:

~ is_empty(bird, cons(bird, B, Bs))).

If the type and the preceding colon (:) are omitted, the variable is given type $i. Every
type variable occurring in a TFF1 formula (whether in a type argument or in the type of
a bound variable) must also be bound, with the pseudotype $tType:



tff(lookup_update_same, axiom,

![A : $tType, B : $tType, M : map(A, B), K : A, V : B]:

lookup(A, B, update(A, B, M, K, V), K) = V).

A single quantifier cluster can bind both type variables and term variables. Universal
and existential quantifiers over type variables are allowed under the propositional con-
nectives, including equivalence, as well as under other quantifiers over type variables,
but not in the scope of a quantifier over a term variable. Rationale: A statement of the
form “for every integer k, there exists a type α such that . . .” effectively makes α a de-
pendent type. On such statements, type skolemization (Section 5.2) is impossible, and
there is no easy translation to ML-style polymorphic formalisms. Moreover, type hand-
ling in an automatic prover would be more difficult were such constructions allowed,
since they require paramodulation into types.

On the other hand, all the notions and procedures described in this specification—
except for type skolemization—are independent of this restriction. The rules of type
checking and the notion of interpretation are directly applicable to unrestricted for-
mulas. The encoding into a monomorphic logic (Section 4) is sound and complete on
unrestricted formulas, and the proofs require no adjustments. This prepares the ground
for TFF2, which is expected to lift the restriction and support more elaborate forms
of dependent types. Implementations of TFF1 are encouraged to support unrestricted
formulas, treating them according to the semantics given here, if practicable.

Terms and Formulas. Apart from the differences described above, the terms and for-
mulas of TFF1 are identical to those of TFF0, as defined in the TFF0 specification [30].

Example. The following problem gives the general flavor of TFF1. It first declares and
axiomatizes lookup and update operations on maps, then conjectures that update is
idempotent for fixed keys and values. Its SZS status [26] is Theorem.

tff(map, type, map : ($tType * $tType) > $tType).

tff(lookup, type,

lookup : !>[A : $tType, B : $tType]: ((map(A, B) * A) > B)).

tff(update, type,

update : !>[A : $tType, B : $tType]:

((map(A, B) * A * B) > map(A, B))).

tff(lookup_update_same, axiom,

![A : $tType, B : $tType, M : map(A, B), K : A, V : B]:

lookup(A, B, update(A, B, M, K, V), K) = V).

tff(lookup_update_diff, axiom,

![A : $tType, B : $tType, M : map(A, B), V : B, K : A, L : A]:

(K != L => lookup(A, B, update(A, B, M, K, V), L) =

lookup(A, B, M, L))).

tff(map_ext, axiom,

![A : $tType, B : $tType, M : map(A, B), N : map(A, B)]:

((![K : A]: lookup(A, B, M, K) = lookup(A, B, N, K)) =>

M = N)).



tff(update_idem, conjecture,

![A : $tType, B : $tType, M : map(A, B), K : A, V : B]:

update(A, B, update(A, B, M, K, V), K, V) =

update(A, B, M, K, V)).

3 Type Checking and Semantics

Notation. Starting with this section, we use standard mathematical notation to write
types, terms, and formulas. Our conventions for metavariables are summarized below.

Type variables: α, β Function symbols: f , g
Type constructors: κ Predicate symbols: p, q
Types: σ, τ Terms: s, t
Term variables: u, v Formulas: ϕ, ψ

Possibly empty lists of types and terms are denoted by σ̄ and t̄, respectively.
We use the symbols ×, →, and ∀ to write type signatures, and write o (lowercase

omicron) for the Boolean pseudotype $o. It is convenient to treat equality (≈), negation
(¬), conjunction (∧), and universal quantification (∀) as logical symbols and regard
disequality (6≈), disjunction (∨), implication (→), reverse implication (←), equivalence
(←→), inequivalence ( 6←→), and existential quantification (∃) as abbreviations. Equality
could be seen as a polymorphic predicate with the type signature ∀α. α×α→ o, but the
type instance is implicitly specified by the type of either argument, instead of explicitly
via a type argument; hence, it is preferable to treat it as a logical symbol.

The set of type variables occurring freely in a formula ϕ (in the type arguments of
polymorphic symbols or in the types of bound variables) is denoted by FVT(ϕ). The set
of free term variables of ϕ is denoted by FV(ϕ). The formula ϕ is closed if both FVT(ϕ)
and FV(ϕ) are empty.

A type substitution ρ is a mapping of type variables to types. A monomorphic type
substitution maps every type variable either to itself or to a monomorphic type. Given
a function h, the expression h[x 7→ a] denotes the function that maps x to a and every
other element y in h’s domain to h(y).

Type Checking. Let γ be a type context, a function that maps every variable to a type.
A type judgment γ ` t : τ expresses that the term t is well-typed and has type τ in
context γ. A type judgment γ ` ϕ : o expresses that the formula ϕ is well-typed in γ.
We write f : ∀α1 . . .αm. τ1×·· ·×τn→ τ and p : ∀α1 . . .αm. τ1×·· ·×τn→ o to specify
type signatures of function and predicate symbols, where m and n can be 0.

The typing rules of TFF1 are given below:

γ ` u : γ(u)

f : ∀α1 . . .αm. τ1×·· ·×τn→ τ γ ` t1 : τ1 ρ · · · γ ` tn : τn ρ

γ ` f (α1 ρ, . . . , αm ρ, t1, . . . , tn) : τρ

p : ∀α1 . . .αm. τ1×·· ·×τn→ o γ ` t1 : τ1 ρ · · · γ ` tn : τn ρ

γ ` p(α1 ρ, . . . , αm ρ, t1, . . . , tn) : o



γ ` s : τ γ ` t : τ
γ ` s≈ t : o

γ ` ϕ : o γ ` ψ : o
γ ` ϕ ∧ ψ : o

γ ` ϕ : o
γ ` ¬ϕ : o

γ[u 7→ τ] ` ϕ : o
γ ` ∀u :τ. ϕ : o

γ ` ϕ[α′/α] : o
γ ` ∀α. ϕ : o

In the last rule, α′ is an arbitrary type variable that occurs neither in ϕ nor in the
values of γ. The renaming is necessary to reject formulas such as ∀α.∀u : α.∀α.∀v : α.
u ≈ v, where the types of u and v are actually different. To simplify the subsequent
definitions, we assume from now on that no type variable can be both free and bound
in the same formula; we call this the no-clash assumption. As a result, we can avoid
explicit renaming of type variables, and the last typing rule’s premise becomes γ ` ϕ : o.

Since every type variable in a polymorphic type signature must be bound in it, it is
impossible for a term to have two different types in the same context.

A closed TFF1 formula ϕ is well-typed iff the judgment γ ` ϕ : o is derivable for
any γ. Obviously, if a closed formula is well-typed in one type context, it is well-typed
in any other one; hence, we omit γ and write ` ϕ : o. Closed well-typed formulas are
called sentences.

Semantics. An interpretation I for a given set of type constructors, function symbols,
and predicate symbols is constructed as follows. First, we fix a nonempty collection D
of nonempty sets, the domains. The union of all domains is called the universe, U.

An n-ary type constructor κ is interpreted as a function κI : Dn→D. Let θ be a type
valuation, a function that maps every type variable to a domain. Types are evaluated
according to the following equations:

JαKIθ , θ(α) Jκ(τ1, . . . , τn)KIθ , κI
(
Jτ1KIθ , . . . , JτnKIθ

)
Since type evaluation depends only on the values of θ on the type variables occurring
in a type, we write JτKI to denote the domain of a monomorphic type τ. We use the
notation [α1 7→ D1, . . . , αm 7→ Dm] to specify θ for types whose free type variables are
among α1, . . . ,αm.

A predicate symbol p : ∀α1 . . .αm. τ1×·· ·×τn→ o is interpreted as a relation pI ⊆
Dm×Un. A function symbol f : ∀α1 . . .αm. τ1×·· ·×τn→ τ is interpreted as a function
fI on Dm×Un that maps any m domains D1, . . . ,Dm and n universe elements to an
element of JτKIθ , where θ maps each αi to Di.

Let ξ be a variable valuation, a function that assigns to every variable an element
of U. TFF1 terms and formulas are evaluated according to the following equations:

JuKIθ,ξ , ξ(u) J¬ϕKIθ,ξ , ¬JϕKIθ,ξ
J f (σ̄, t̄ )KIθ,ξ , fI

(
Jσ̄KIθ ,Jt̄ KIθ,ξ

)
Jϕ ∧ ψKIθ,ξ , JϕKIθ,ξ ∧ JψKIθ,ξ

Jp(σ̄, t̄ )KIθ,ξ , pI
(
Jσ̄KIθ ,Jt̄ KIθ,ξ

)
J∀u : τ. ϕKIθ,ξ , ∀a∈ JτKIθ . JϕK

I
θ,ξ[u 7→a]

Jt1 ≈ t2KIθ,ξ ,
(
Jt1KIθ,ξ = Jt2KIθ,ξ

)
J∀α. ϕKIθ,ξ , ∀D ∈ D. JϕKIθ[α 7→D], ξ

We omit irrelevant subscripts and write JϕKI to denote the evaluation of a sentence.



A sentence ϕ is true in an interpretation I, written I |= ϕ, iff JϕKI is true. The
interpretation I is then a model of ϕ. A sentence that has a model is satisfiable. A
sentence that is true in every interpretation is valid. These notions are extended to sets
and sequents of TFF1 sentences in the usual way.

4 Translation to TFF0

We describe a simple translation from TFF1 to the classical many-sorted first-order
logic TFF0. The translation is included here for illustrative purposes; more suitable
encoding schemes for applications are discussed in Section 7.

Our strategy for translating types is to encode them as terms and use a special binary
predicate to encode type information. To avoid mixing types and terms in the encoded
problem, we introduce two sorts, D and U, corresponding to the set of domains D and
the universe U, respectively.1

Let ∆ be a set of TFF1 sentences. We construct an equisatisfiable set of mono-
morphic two-sorted formulas M (∆) as follows. To every type variable α in ∆ or in
the type signature of a function symbol, we assign a fresh variable α̂ of sort D. To
every term variable u, we assign a fresh variable û of sort U. To every n-ary type
constructor κ, we assign a function symbol κ̂ : Dn → D. To every function symbol
f : ∀α1 . . .αm. τ1× ·· · × τn → τ, we assign a function symbol f̂ : Dm×Un → U. To
every predicate symbol p : ∀α1 . . .αm. τ1×·· ·× τn→ o, we assign a predicate symbol
p̂ : Dm×Un→ o. Finally, we introduce the special predicate symbol Ty : U×D→ o.

The M transformation translates TFF1 types, terms, and formulas according to the
following equations:

M (α) , α̂ M (κ(σ̄)) , κ̂(M (σ̄))

M (u) , û M (¬ϕ) , ¬M (ϕ)

M ( f (σ̄, t̄ )) , f̂ (M (σ̄),M (t̄ )) M (ϕ ∧ ψ) , M (ϕ) ∧M (ψ)

M (p(σ̄, t̄ )) , p̂(M (σ̄),M (t̄ )) M (∀u : τ. ϕ) , ∀û. Ty(û,M (τ))→M (ϕ)

M (t1 ≈ t2) , M (t1)≈M (t2) M (∀α. ϕ) , ∀α̂.M (ϕ)

The M transformation is lifted to lists of types or terms in the evident, pointwise way.
The inhabitation (or nonemptiness) axiom, INH, is the formula ∀α̂.∃û. Ty(û, α̂). For

each function symbol f : ∀α1 . . .αm. τ1× ·· ·× τn → τ, the associated typing axiom is
the formula

∀α̂1 . . . α̂m. ∀û1 . . . ûn. Ty( f̂ (α̂1, . . . , α̂m, û1, . . . , ûn),M (τ))

We let AX∆ denote the set of typing axioms associated with all function symbols occur-
ring in ∆. Finally, we define

M (∆) , {M (ϕ) | ϕ ∈ ∆} ∪ AX∆ ∪ {INH}
1 In fact, because the generated TFF0 problems are monotonic, the sorts can be safely erased to

produce an equisatisfiable FOF problem [9].



It is easy to see that M converts TFF1 types into well-formed TFF0 terms of sort D,
TFF1 terms into well-formed TFF0 terms of sort U, and TFF1 formulas into well-for-
med TFF0 formulas.

Theorem 1 (Soundness of MM ). If a set of sentences ∆ is satisfiable in TFF1, then
M (∆) is satisfiable in TFF0.

Proof. Let M be a model of ∆. We construct an interpretation I of M (∆) as follows.
Let D and U stand for the set of domains and the universe in M, respectively. In I,
we define the domain of sort D to be D and the domain of sort U to be U. Each type
constructor κ̂, function symbol f̂ , or predicate symbol p̂ is interpreted in I exactly as
the type constructor κ, function symbol f , or predicate symbol p in M. The predicate
symbol Ty is interpreted as the membership relation.

We show that for every type σ, term t, and formula ϕ occurring in ∆, and for every
variable valuation ν in I, we have

JM (σ)KIν = JσKMθ JM (t)KIν = JtKMθ,ξ JM (ϕ)KIν = JϕKMθ,ξ

where θ(α), ν(α̂) and ξ(u), ν(û) for every type variable α and every variable u. We
prove these equalities by induction on the structure of types, terms, and formulas. The
only nontrivial case is that of a quantified formula, ∀u : τ. ϕ. It is easy to see that any
a ∈ U belongs to JτKMθ iff JTy(û,M (τ))KIν′ , where ν′ = ν[û 7→ a]. Indeed, the latter is
equivalent to a∈ JM (τ)KIν′ . Since ν and ν′ produce the same θ in M, JM (τ)KIν′ = JτKMθ .
Then J∀û. Ty(û,M (τ))→M (ϕ)KIν is exactly J∀u : τ. ϕKMθ,ξ by induction hypothesis.

Now we must only ensure that the typing axioms AX∆ and the inhabitation axiom
INH hold in I. This immediately follows from the definition of interpretation in TFF1
and the first equality above. ut

Theorem 2 (Completeness of MM ). Given a set of sentences ∆, if M (∆) is satisfiable
in TFF0, then ∆ is satisfiable in TFF1.

Proof. Let M0 be a model of M (∆). We first construct a model M of M (∆) from M0
by replacing every element d in the domain of D by the set D= {〈a, d〉 | ∃a.TyM0(a, d)}
and updating the interpretations of type constructors and function and predicate sym-
bols in M0 accordingly. We can safely perform this substitution: Since the inhabitation
axiom INH holds in M0, every set D is nonempty, and distinct elements are mapped to
distinct sets. The predicate TyM(a, D) holds iff D contains a pair 〈a, d〉 for some d.

We construct an interpretation I of ∆ as follows. The set of domains D is the domain
of D in M. As usual, U denotes the union of all domains in D. Note that some elements
of the domain of U may not appear in any pair in U. The symbols of ∆ are interpreted
in I according to the equations

κI(D1, . . . ,Dm) , κ̂M(D1, . . . ,Dm)

fI(D1, . . . ,Dm,〈a1,d1〉, . . . ,〈an,dn〉) ,
〈

f̂ M(D1, . . . ,Dm,a1, . . . ,an), d
〉

pI(D1, . . . ,Dm,〈a1,d1〉, . . . ,〈an,dn〉) , p̂M(D1, . . . ,Dm,a1, . . . ,an)



where κ is of arity m, p : ∀α1 . . .αm. τ1×·· ·×τn→ o, f : ∀α1 . . .αm. τ1×·· ·×τn→ τ,
and d is the fixed second coordinate of pairs in the domain

D = JM (τ)KM[α̂1 7→D1,...,α̂m 7→Dm]
= JτKI[α1 7→D1,...,αm 7→Dm]

Since the typing axioms AX∆ hold in M, we have TyM
(

fM(D1, . . . ,Dn,a1, . . . ,am),D
)
,

and therefore the result of fI indeed belongs to D.
Given a type context γ, a type valuation θ, and a variable valuation ξ, we say that

they are admissible for a TFF1 formula ϕ if the following conditions are satisfied:

• ϕ is well-typed in context γ;
• for every variable v free in ϕ, no type variable occurring in γ(v) is bound in ϕ;
• for every variable v free in ϕ, we have ξ(v) ∈ Jγ(v)KIθ .

Obviously, any triple γ, θ, ξ is admissible for a sentence. We must show that for
every type σ, term t, and formula ϕ, and for all γ, θ, ξ admissible for ϕ, we have

JσKIθ = JM (σ)KMν π1(JtKIθ,ξ) = JM (t)KMν JϕKIθ,ξ = JM (ϕ)KMν

where π1 stands for the first projection of a pair, ν(α̂) , θ(α), and ν(û) , π1(ξ(u)) for
every type variable α and variable u.

The proof is by induction on the structure of types, terms, and formulas. There are
three nontrivial cases: equality, quantification over a term variable, and quantification
over a type variable.

Let ϕ be an equality t1 ≈ t2. Let 〈a1, d1〉 be Jt1KIθ,ξ and 〈a2, d2〉 be Jt2KIθ,ξ. By in-
duction hypothesis, JM (t1)KMν = a1 and JM (t2)KMν = a2. We must show d1 = d2. By
assumption, ϕ is well-typed in γ. Thus, γ ` t1 : σ and γ ` t2 : σ for some type σ. If
t1 is a variable v, then Jt1KIθ,ξ = ξ(v) ∈ Jγ(v)KIθ,ξ = JσKIθ . Otherwise, t1 is a function ap-
plication f (σ1, . . . ,σm, s1, . . . , sn), for f : ∀α1 . . .αm. τ1×·· ·× τn → τ. Let Di = JσiKIθ
for every i ∈ [1,m]. Then σ= τ[σ1/α1, . . . ,σn/αn] and JσKIθ = JτKI[α1 7→D1,...,αn 7→Dn]

. By

construction of I, Jt1KIθ,ξ ∈ JσKIθ . By the same argument, Jt2KIθ,ξ ∈ JσKIθ . Hence, d1 = d2

and JϕKIθ,ξ = JM (ϕ)KMν .
Let ϕ be a quantified formula ∀u : τ. ψ. We must show that for every pair 〈a, d〉 in

JτKIθ , we have TyM(a,JM (τ)KMν′ ), where ν′ = ν[û 7→ a], and vice versa, for every a in
the domain of U, if TyM(a,JM (τ)KMν′ ) holds, then there exists some d such that 〈a, d〉 ∈
JτKIθ . Since term variables do not occur in types, JM (τ)KMν′ = JτKIθ . By construction of
M, TyM(a,JτKIθ ) holds iff there exists some d′ such that 〈a, d′〉 ∈ JτKIθ . Now, notice
that the triple γ[u 7→ τ], θ, ξ[u 7→ 〈a, d〉] is admissible for ψ. Indeed, ψ is well-typed in
γ[u 7→ τ], and for every v ∈ FV(ψ), we have

ξ[u 7→ 〈a, d〉](v) ∈ Jγ[u 7→ τ](v)KIθ

Also, if τ contains a type variable α bound in ψ, then α is both free and bound in ϕ,
which violates the no-clash assumption. Then, by induction hypothesis, J∀u : τ. ψKIθ,ξ is
exactly J∀û. Ty(û,M (τ))→M (ψ)KMν .

Let ϕ be a quantified formula ∀α. ψ. Let D be an element of D. Under the no-clash
assumption, formula ψ is well-typed in γ, and for all v ∈ FV(ψ) = FV(ϕ), we have
Jγ(v)KIθ[α 7→D] = Jγ(v)KIθ . Then JϕKIθ,ξ = JM (ϕ)KMν by induction hypothesis. ut



5 Preprocessing of Type Quantifiers

We describe two preprocessing steps that preserve both satisfiability and unsatisfiabil-
ity of TFF1 problems. They eliminate phantom type variables in type signatures and
alternating ∀/∃ type quantifier prefixes in formulas, two features that are not directly
supported by ML-style formalisms (which are otherwise a good match for TFF1).

5.1 Elimination of Phantom Type Variables

ML-style formalisms, as implemented in Alt-Ergo [6], Boogie [17], HOL [12], HOL
Light [13], Isabelle/HOL [20], Why3 [8], and several other systems, allow type vari-
ables in type signatures, but without explicit ∀-binders. Instead of relying on explicit
type arguments, these systems determine the concrete instance of a function or predi-
cate symbol’s type signature by the types of its term arguments, the type of the result,
and optional type annotations inside terms. The natural translation from TFF1 to such a
formalism would map ∀αβ. α×β→ o (and ∀βα. α×β→ o) to α×β→ o, simply omit-
ting the ∀-binders. To compensate for the missing type arguments, type annotations are
sometimes needed to guide the Hindley–Milner type inference.

The difficulties arise in conjunction with phantom type variables: If ∀α. o collapses
to o, the dependency on the type is lost. To ease the adoption of TFF1 in such systems,
we suggest a preprocessing step that eliminates phantom type variables. This step is
lightweight, as it requires only the introduction of one term argument per phantom type.
In particular, it is the identity for formulas that do not rely on phantom type variables.

Let ∆ be a set of sentences. We construct an equisatisfiable set E (∆) as follows.
We introduce a special unary type constructor Ph. We replace every function symbol
f : ∀α1 . . .αm. τ1× ·· ·× τn → τ with a new function symbol f̂ : ∀α1 . . .αm. Ph(αi1)×
·· ·×Ph(αir)×τ1×·· ·×τn→ τ, where αi1 , . . . ,αir are the type variables from the quan-
tifier prefix that do not occur in τ1, . . . , τn or τ. Every predicate symbol p : ∀α1 . . .αm.
τ1× ·· · × τn → o is replaced with a new predicate symbol p̂ : ∀α1 . . .αm. Ph(αi1)×
·· · × Ph(αir)× τ1 × ·· · × τn → o, where αi1 , . . . ,αir are the type variables from the
quantifier prefix that do not occur in τ1, . . . , τn. Finally, we add a “witness” constant
Wt : ∀α. Ph(α). The E transformation translates terms and atomic formulas as follows:

E (u) , u E (s≈ t) , E (s)≈ E (t)

E ( f (σ1, . . . ,σm, t̄ )) , f̂ (σ1, . . . ,σm,Wt(σi1), . . . ,Wt(σir),E (t̄ ))

E (p(σ1, . . . ,σm, t̄ )) , p̂(σ1, . . . ,σm,Wt(σi1), . . . ,Wt(σir),E (t̄ ))

Given a formula ϕ or a set of sentences ∆, E (ϕ) and E (∆) denote the result of applying
E to every atomic formula in ϕ and ∆, respectively.

Theorem 3. Any ∆ is equisatisfiable to E (∆).

Proof. A model of ∆ can be converted to a model of E (∆) as follows. We choose
an arbitrary value e and take the singleton {e} as the domain of Ph(D) for any do-
main D. Accordingly, Wt evaluates to e on any argument. Interpretations of other sym-
bols are adjusted in the obvious way. Conversely, we evaluate any function symbol f
on D1, . . . ,Dm,a1, . . . ,an in ∆ exactly as f̂ on D1, . . . ,Dm,e1, . . . ,er,a1, . . . ,an in E (∆),
where each ek is the evaluation of Wt on Dik , and similarly for predicate symbols. ut



5.2 Type Skolemization

Another feature of TFF1 that is not universally supported by systems with polymorphic
types is explicit quantification over type variables. In particular, HOL and other ML-
style logics consider all type variables implicitly universally quantified at the top of a
formula. The solution is type skolemization. We define two transformations, S + and
S −, to perform type skolemization in axioms and in conjectures, respectively:

S +(p(σ̄, t̄ )) , p(σ̄, t̄ ) S −(p(σ̄, t̄ )) , p(σ̄, t̄ )

S +(t1 ≈ t2) , t1 ≈ t2 S −(t1 ≈ t2) , t1 ≈ t2

S +(¬ϕ) , ¬S −(ϕ) S −(¬ϕ) , ¬S +(ϕ)

S +(ϕ ∧ ψ) , S +(ϕ) ∧S +(ψ) S −(ϕ ∧ ψ) , S −(ϕ) ∧S −(ψ)

S +(∀u : τ. ϕ) , ∀u : τ. ϕ S −(∀u : τ. ϕ) , ∀u : τ. ϕ

S +(∀α. ϕ) , ∀α.S +(ϕ) S −(∀α. ϕ) , S −(ϕ[κ(β̄)/α])

where κ is a fresh type constructor and β̄ is the list of free type variables of ∀α. ϕ. Since
TFF1 forbids quantifiers over type variables in the scope of a quantifier over a term
variable, S + and S − simply stop at the outermost term quantifier.

To simplify the presentation, we viewed equivalence (←→) and inequivalence ( 6←→) as
abbreviations. A slight complication arises in practice when processing type quantifiers
under either connective, where they are unpolarized. The easiest solution is to expand
the connective before skolemizing type quantifiers that appear under it.

Given a set of TFF1 sentences ∆, S +(∆) and S −(∆) denote the result of applying
the corresponding transformations to every formula in ∆.

Theorem 4. Any ∆ is equisatisfiable to S +(∆).

Proof. It is easy to see that skolemizing the D-sorted variables in M (∆) gives exactly
M (S +(∆)) modulo renaming of Skolem symbols and permutation of their arguments.
Theorems 1 and 2 conclude the proof. ut

6 Applications

A number of applications already support TFF1. Geoff Sutcliffe has extended the TPTP
World infrastructure to process TFF1 problems and solutions. This involved adapting
the Backus–Naur form specification of the TPTP syntaxes, from which parsers are gen-
erated.2 Some TPTP tools still need to be ported to TFF1; this is ongoing work.

The Why3 [8] environment, which defines its own ML-like polymorphic specifica-
tion language, can parse pure TFF1. Why3 translates between TFF1 and a wide range of
formats, including FOF, SMT-LIB, and Alt-Ergo’s native syntax. In addition, Why3’s
TFF1 parser is being ported to Alt-Ergo [6], so that it can directly process TFF1.

HOL(y)Hammer [14] and Sledgehammer [21] integrate various automatic provers
in the proof assistants HOL Light and Isabelle/HOL. They have been extended to output

2 http://www.cs.miami.edu/~tptp/TPTP/SyntaxBNF.html



pure TFF1 problems for Alt-Ergo and Why3 (in addition to FOF, TFF0, and THF0).
Using Sledgehammer, we produced 987 problems to populate the TPTP library.3

7 Related Work

Formalisms. The TPTP family of formats is well established in the automated reasoning
community. As part of the MPTP project [31], Urban designed private extensions of the
TPTP FOF syntax with dependent types to accommodate Mizar, as well as translations;
these could form the basis of a future TFF2 format. The full THF syntax [29], which is
not yet finalized or implemented, also supports dependent types.

For interactive theorem provers, polymorphism is the norm rather than the excep-
tion. HOL systems [12,13,20] provide ML-style (rank-1) polymorphism, while Coq [3]
supports dependent types and higher-rank polymorphism. The SMT solver Alt-Ergo [6]
is perhaps the only automatic prover that supports polymorphism natively.

The intermediate verification language and tool Boogie 2 [17] supports a restricted
form of higher-rank polymorphism (due to its polymorphic maps), and its cousin Why3
[8] provides rank-1 polymorphism. Both provide TPTP and SMT-LIB backends.

Encodings. Early descriptions of type encodings are due to Enderton [11, §4.3], Stickel
[24, p. 99], and Wick and McCune [34, §4]. TFF1 type arguments are reminiscent of
System F; a FOF-based encoding that exploits them is described by Meng and Paulson
[19], who also present a translation of axiomatic type classes.

Considerable progress has been made lately toward sound, complete, and efficient
encodings of polymorphic logics in untyped or monomorphic logics. Leino and Rüm-
mer [17] present a translation of higher-rank polymorphism, including explicit quan-
tifiers over types, into the many-sorted SMT-LIB syntax, while preserving interpreted
types. They also show how to exploit SMT triggers to prevent unsound variable in-
stantiations in a translation based on type arguments. Bobot and Paskevich [7] extend
earlier work by Couchot and Lescuyer [10] to encode polymorphism while preserving
arbitrary monomorphic types (e.g., array types). Blanchette et al. [5], building on work
by Claessen et al. [9], present a lightweight encoding of polymorphic types that ex-
ploits type monotonicity. All of these translations assume elimination of phantom type
variables (Section 5.1); the last two also assume type skolemization (Section 5.2).

8 Conclusion

This paper described the TPTP TFF1 format, an extension of the monomorphic TFF0
format with rank-1 polymorphism. The new format complements the existing TPTP
offerings. For reasoning tools that already support polymorphism, TFF1 is a portable
alternative to the existing ad hoc syntaxes. But more importantly, the format is a ve-
hicle to foster native polymorphism support in automatic theorem provers. The time is
ripe: After many years of untyped reasoning, we have recently witnessed the rise of
interpreted arithmetic embedded in monomorphic logics. TFF1 lifts the most obvious
restrictions of such systems.

3 http://www.cs.miami.edu/~tptp/TPTP/Proposals/TFF1.html



TFF1 is part of TPTP World. The TPTP library already contains nearly a thousand
TFF1 problems, and although the format is in its infancy, it is supported by several
applications, including the SMT solver Alt-Ergo (via Why3). Work has commenced in
Saarbrücken to add polymorphism to the superposition prover SPASS [33]. Given that
many applications require polymorphism, other reasoning tools are likely to follow suit.
The annual CADE Automated System Competition (CASC) will certainly have a role
to play driving adoption of the format. But regardless of progress in prover technology,
equipped with a concrete syntax and suitable middleware, users can already turn their
favorite automatic theorem prover into a fairly efficient polymorphic prover.

Rank-1 polymorphism is, of course, no panacea. More advanced features, such as
type classes, higher ranks, and dependent types, are not catered for (although type
classes can be comfortably encoded in TFF1). These are expected to be part of a future
TFF2, with the proviso that there be sufficient interest from users and implementers.

